

Este es el script “guion”
Nota: v23 es el vector de V2 a V3, v34 el vector de V3 a V4, etc.

1. ACTUALIZA LA POSICIÓN DE LOS VÉRTICES ADYACENTES A V1.

```
Valor(V2, Interseca(Semirrecta(V1,V2), Esfera(V1,1)))  
Valor(V3, Interseca(Semirrecta(V1,V3), Esfera(V1,1)))  
Valor(V4, Interseca(Semirrecta(V1,V4), Esfera(V1,1)))  
ActualizaConstrucción()
```

2. TRANSMITE LAS POSICIONES A LOS TRES VÉRTICES SIGUIENTES.

2.1 La mayoría de las veces, V5, V6 y V7 ocuparán las posiciones P1, P2, P3 que se calculan a continuación:

```
Valor(C, (V3+V2)/2)  
Valor(Q1, Interseca(Recta(C, v23), PlanoPerpendicular(V5, v23)))  
Valor(P1, Interseca(Semirrecta(C, C+V5-Q1), Circunferencia(C, abs(C-V1), v23)))  
  
Valor(C, (V4+V3)/2)  
Valor(Q1, Interseca(Recta(C, v34), PlanoPerpendicular(V6, v34)))  
Valor(P2, Interseca(Semirrecta(C, C+V6-Q1), Circunferencia(C, abs(C-V1), v34)))  
  
Valor(C, (V2+V4)/2)  
Valor(Q1, Interseca(Recta(C, v42), PlanoPerpendicular(V7, v42)))  
Valor(P3, Interseca(Semirrecta(C, C+V7-Q1), Circunferencia(C, abs(C-V1), v42)))
```

2.2 Observamos los casos especiales de coincidencia de vértices:

```
Valor(P1, Si(V2==V3, Interseca(Semirrecta(V2,V5), Esfera(V2,1)), P1))  
Valor(P2, Si(V3==V4, Interseca(Semirrecta(V3,V6), Esfera(V3,1)), P2))  
Valor(P3, Si(V4==V2, Interseca(Semirrecta(V4,V7), Esfera(V4,1)), P3))
```

2.3 Si el circunradio r del triángulo P1 P2 P3 es mayor que 1, las posiciones de P1, P2 y P3 deben reajustarse:

```
Valor(C, Centro(Circunferencia(P1,P2,P3)))  
Valor(r, Si(P1==P2, abs(P1-P3)/2, P2==P3, abs(P2-P1)/2, P3==P1, abs(P3-P2)/2, abs(C-P1)))  
Valor(C, Centro(Circunferencia(V2,V3,V4)))  
Valor(P', Interseca(Semirrecta(C,2C-V1), Esfera(C,1)))
```

2.3.1 De los dos puntos de intersección, elige como P1 el más cercano a V5:

```
Valor(C, (V3+V2)/2)
```

Valor(pc, {Interseca(Esfera(P',1), Circunferencia(C, abs(C-V1), v23))})
 Valor(Q1, pc(1))
 Valor(Q2, pc(2))
 Valor(Q1, Si(abs(P1-Q2)<abs(P1-Q1), Q2, Q1))
 Valor(V5, Si(r > 1, Q1, P1))

2.3.2 De los dos puntos de intersección, elige como P2 el más cercano a V6:

Valor(C, (V4+V3)/2)
 Valor(pc, {Interseca(Esfera(P',1), Circunferencia(C, abs(C-V1), v34))})
 Valor(Q1, pc(1))
 Valor(Q2, pc(2))
 Valor(Q1, Si(abs(P2-Q2)<abs(P2-Q1), Q2, Q1))
 Valor(V6, Si(r > 1, Q1, P2))

2.3.3 De los dos puntos de intersección, elige como P3 el más cercano a V7:

Valor(C, (V2+V4)/2)
 Valor(pc, {Interseca(Esfera(P',1), Circunferencia(C, abs(C-V1), v42))})
 Valor(Q1, pc(1))
 Valor(Q2, pc(2))
 Valor(Q1, Si(abs(P3-Q2)<abs(P3-Q1), Q2, Q1))
 Valor(V7, Si(r > 1, Q1, P3))

3. DETERMINA LA POSICIÓN DE V8 (EL VÉRTICE OPUESTO A V1).

3.1 Primer caso. V5, V6 y V7 diferentes (si son diferentes ya no pueden estar alineados):

Valor(C, Centro(Circunferencia(V5,V6,V7)))
 Valor(v, VectorNormal(Plano(V5,V6,V7)))
 Valor(pc, {Interseca(Recta(C, v), Esfera(V5,1))})
 Valor(Q1, pc(1))
 Valor(Q2, pc(2))
 Valor(Q1, Si(abs(V8-Q2)<abs(V8-Q1), Q2, Q1))
 Valor(V8, Si(V5!=V6 && V5!=V7 && V7!=V6, Q1, V8))

3.2 Segundo caso. V5 coincide con V6 pero no con V7:

Valor(C, (V5+V7)/2)
 Valor(Q1, Interseca(Recta(C, v57), PlanoPerpendicular(V8, v57)))
 Valor(P1, Interseca(Semirrecta(C, C+V8-Q1), Circunferencia(C, abs(C-V2), v57)))
 Valor(V8, Si(V5==V6 && V5!=V7, P1, V8))

3.3 Tercer caso. V7 coincide con V5 pero no con V6, o V7 coincide con V6 pero no con V5:

```
Valor(C, (V5+V6)/2)
Valor(Q1, Interseca(Recta(C, v56), PlanoPerpendicular(V8, v56)))
Valor(P1, Interseca(Semirrecta(C, C+V8-Q1), Circunferencia(C, abs(C-V3), v56)))
Valor(V8, Si((V7==V5 || V7==V6) && V5!=V6, P1, V8))
```

3.4 Cuarto caso. Coincidén V5, V6 y V7:

```
Valor(V8, Si(V5==V6 && V5==V7, Interseca(Semirrecta(V5,V8), Esfera(V5,1)), V8))
ActualizaConstrucción()
```